
GENERAL DESCRIPTION

The LocalTalk Link Access Protocol (LLAP) is the ISO-OSI
link layer protocol of the AppleTalk network system using
LocalTalk. Along with ELAP (the corresponding Ethernet
link layer protocol) and TLAP (the Token Ring link layer
protocol), it provides the foundations upon which the other
protocols rest. The LLAP protocol supports the node-to-
node transmission of packets used by DDP and RTMP to
route packets around the internetwork; DDP, in turn,
supports the name binding functions of NBP, the reliable
frame delivery of ATP, and the rest of the AppleTalk
protocol stack.

A majority of the difficult timing and all of the hardware
interface problems crop up in the LLAP driver. These
problems are so difficult that it makes sense to start writing
such a driver by writing experimental routines that transmit
and receive frames. This App Note addresses the intricacies
of the interframe and interdialog timings before trying to
engineer code that will truly be a driver. Also, some of the
experimental routines to run on the Z80181 Emulation
Adapter Board will be explained.

INTRODUCTION

The LLAP (LocalTalk Link Access Protocol) is the ISO/OSI
(International Standards Organization/Open Systems
Interconnection) link layer protocol of the AppleTalk network
system. This protocol manages the node-to-node
transmission of data packets in the network. LLAP governs
access to the link and provides a means for nodes to
discover valid addresses. It does not guarantee packet
delivery; it does guarantee that those packets that are
delivered are error-free.

This Appnote (Application Note) does not address the
architectural issues of writing a driver but it does focus on
the details of using an SCC to send and receive LLAP
frames. However, some of the problems of transmitting
and receiving LLAP frames are discussed, using sample
code written for Zilog’s Z80181 Emulation Adapter Board.
Also, the problems of sending sync pulses, timing
transmissions and determining that a frame has been
received properly will be discussed.

he LLAP Protocol is an important part of the Appletalk network system. It manages access
to the node-to-node transmission of network data packets, governs access to the link, and
provides a means for nodes to discover valid addresses...all error free.T

GENERAL DESCRIPTION (Continued)

The LLAP provides the basic transmission of packets from
one node to another on the same network. LLAP accepts
packets of data from clients residing on a particular node
and encapsulates that data into its proper LLAP data
packet. The encapsulation includes source and destination
addresses for proper delivery. LLAP ensures that any
damaged packet is discarded and rejected by the
destination node. The LLAP makes no effort to deliver
damaged packets.

Carrier Sense Multiple Access with
Collision Avoidance

It is LLAP’s responsibility to provide proper link access
management to ensure fair access to the link by all nodes
on that network. The access discipline that governs this is
known as Carrier Sense Multiple Access with Collision
Avoidance (CSMA/CA). A node wishing to gain access to
the link must first sense that the link is not in use by any
other node (carrier sense); if the link has activity, then the
node wishing to transmit must defer transmission. The
ability of LLAP to allow multiple access to the link also
leaves room for possible collisions with other data packets.
LLAP attempts to minimize this probability (collision
avoidance).

Two techniques are used by LLAP in its implementations
of CSMA/CA. LLAP outlines this procedure but falls short
in endorsing which hardware to use. (The SCC is, of
course, used by Apple.) The first technique takes
advantage of the distinctive 01111110 flag bytes that
encapsulate the data packet (note that this implies that
SDLC is used). LLAP stipulates that a minimum of two flags
precede each of these data packets. The leading flag
characters provide byte synchronization and give a clue
to any listener that some other node is using the link at a
particular time (use the Hunt bit in RR0 if the SCC is used).

In SDLC mode, the receiver automatically synchronizes
on the flag byte and resets the Hunt bit to zero. The SCC
has some latency in detecting these flag bytes due to the
shifter, etc. This is not ideal because the node needing to
transmit may determine that the link is free, when in fact the
flag bytes are still being shifted into its receiver (i.e., the
link is not idle at all).

A closing flag is also needed to fully encapsulate the data
packet. LLAP requires that 12 to 18 ones be sent after this
closing flag. The LocalTalk hardware (i.e., the SCC)
interprets this as an abort sequence and causes the
node’s hardware to lose byte sync; this then confirms that
the current sender’s transmission is over. In SDLC mode,
seven or more contiguous 1’s in the receive data stream
forces the receiver into Hunt (Hunt bit set) and an External/
Status interrupt can be generated. This is important
because the node wishing to use the bus can simply wait
for this interrupt before preparing to transmit it’s packet.

LLAP uses a second technique in its carrier sensing. LLAP
requires that a synchronization pulse for an idle period of
at least two bit times be transmitted prior to sending the
RTS handshaking frame (Figure 1). This synchronization is
obtained by first enabling the hardware line so that an
edge is detected by all the receivers on the network. This
initial edge is perceived as the beginning of the clocking
period. It is soon followed by an idle period (a period with
no carrier) of at least two bit times. All the receivers on the
network see this idle period and assume that the clock has
been lost (missing clock bit set on RR10 ). This method is
much more immediate than the byte flag synchronization
method and provides a quicker way of determining whether
the link is in use. Unfortunately, an interrupt is not generated
by this missing clock and, therefore, polling must be
implemented.

The Z80181 code used for polling the missing clock bit is
approximately fifty clock cycles which at 10 MHz is about
5 µsec or about one bit time. This is still relatively quicker
than the time required for the SCC to reset the Hunt bit (the
flag character takes at least eight bit times for it to be
shifted through the buffer before the Hunt bit is reset to
zero). Synchronization pulses can be sent before every
frame but because of the time constraints associated with
the interframe gaps it makes sense to send such pulses
only before the lapENQ and lapRTS frames.

Figure 1.  CSMA/CA Synchronization Pulse Timing Diagram

Dynamic Node ID

LLAP requires the use of an 8-bit node identifier number
(node ID) for each node on the link. Apple had decided
that all LLAP nodes must have a dynamically assigned
node ID. A node would assign itself its unique address
upon activation. It is then up to that particular node to
ascertain that the address it had chosen is unique. A node
randomly chooses an 8-bit address (for example, the
refresh register value on the Z80181 is added to a randomly
chosen value on the receive buffer to obtain a pseudo
random 8-bit address).

The node then sends out an LLAP Enquiry control packet
to all the other nodes and waits for the prescribed interframe
gap of 200 µsec. If another node is already using this node
ID, then that node must respond within 200 µsec with a
LLAP Acknowledgment control packet. The new node
must then rebroadcast a new guess for its node ID. If a
LLAP Acknowledgment packet is not received within 200
µsec then the new node assumes that the address is
indeed unique. The new node must rebroadcast the LLAP
enquiry packet several more times to account for cases
when the packet could have been lost or when the guessed
node ID is busy and could have missed the Enquiry
packet.

LLAP Packet

LLAP packets are made up of three header bytes
(destination ID, source ID and LLAP type) and 0 to 600
bytes of variable length data. The LLAP type indicates the
type of packet that is being sent. 80H to FFH are reserved
as LLAP control packets. The four LLAP control packets
that are currently being used are: The lapENQ, which is

used as enquiry packet for dynamic node assignments;
the lapACK, which is the acknowledgment to the lapENQ;
the lapRTS, which is the request to send packet that
notifies the destination of a pending transmission; and the
lapCTS, which is the clear-to-send packet in response to
the RTS packet. Control packets do not contain data fields.

LLAP Packet Transmission

LLAP distinguishes between two types of transmissions: a
directed packet is sent from the source node to a specific
destination node through a directed transmission dialog; a
broadcast packet is sent from the source node to all nodes
on the link (destination ID is FFH) through a broadcast
transmission dialog. All dialogs must be separated by a
minimum Inter Dialog Gap (IDG) of 400 µsec. Frames
within these dialogs must be separated from each other
with a maximum Inter Frame Gap (IFG) of 200 µsec.

The source node uses the physical layer to detect the
presence or the absence of data packets on the link. The
node will wait until the line is no longer busy before
attempting to send its packets. If the node senses that the
line is indeed busy, then this node must defer. When the
node senses that the line is idle, then the node waits the
minimum IDG plus some randomly generated time before
sending the packet (the line must remain idle throughout
this period before attempting to send the packet). The
initial packets to be sent are handshaking packets. The
first packet sent by the source node to its destination node
is the RTS packet. The receiver of this RTS packet must
return a CTS packet within the allowable maximum IFG.
The source node then starts transmitting the rest of its data
packet upon receiving this CTS.

Partial Flag Flag LLAP Packet CRC CRCFlag 12-18 1's

TxUnderrun Int.

GENERAL DESCRIPTION (Continued)

Collisions are more likely to occur during the handshaking
phase of the dialog. The randomly generated time that is
added to the IDG tends to help spread out the use of the
link among all the transmitters. A successful RTS to CTS
handshake signifies that a collision did not take place. An
RTS packet that collides with another frame has corrupt
data that shows up as a CRC error on the receiving or the
destination node. Upon receiving this, the destination
node infers that a collision must have taken place and
abstains from sending its CTS packet. The source or the
transmitting node sees that the CTS packet was not received
during the IFG and also infers that a collision did take
place. The sending node then backs off and retries.

The LLAP keeps two history bytes that log the number of
deferrals and collisions during a dialog. These history
bytes help determine the randomly generated time that is
added to the IDG. The randomly generated time is

HARDWARE CONFIGURATION

As shown in Figure  2, the hardware used to implement this
LLAP driver consists of the Z80181 (an integration of the
Z80180 compatible MPU core with one channel of a
Z85C30 SCC, Z80 CTC, two 8-bit general-purpose parallel
ports and two chip select signals) operating at 10 MHz, a
3.6864 MHz clock source and an RS-422 line driver with tri-
state.

The SCC’s clocking scheme decouples the micro-
processor’s clock from the communication clock (Figure
3). The DPLL uses the /RTxC pin as its source. The /RTxC
also drives the Baud Rate Generator which divides its input
by sixteen. The resulting 230.4 kHz signal is then used as

readjusted according to the traffic conditions that are
present on the link. If collisions or deferrals have just
occurred on the most recently sent packets, then it can be
assumed that the link has heavier than usual traffic. Here,
the randomly generated number should be a larger number
in order to help spread out the transmission attempts.
Similarly, if the traffic is not so great, then the randomly
generated number should be smaller, thus reducing the
dispersion of the transmission attempts.

LocalTalk Physical Layer

LocalTalk uses the SDLC format and the FM0 bit encoding
technique. The RS-422 signalling standard for transmission
and reception was chosen over the RS-232 because a
higher data rate over a longer physical distance is required.
LocalTalk requires signals at 230.4 Kbits per second over
a distance of 300 meters.

transmitter clock. This 230.4 kHz signal is also used by one
of the Z80181’s counter/timer trigger inputs (Z80 CTC’s
channel 1) which is used to count the number of elapsed
bit times. In counter mode, each active edge to the CTC’s
CLK/TRG1 input causes the downcounter of the CTC to be
decremented. The /TRxC pin is programmed as BRG
output and is connected to the CLK/TRG1 input through an
external wire.

The /RTS signal is used to tri-state RS-422 to allow other
node transmitters to drive the line. This signal is asserted
and deasserted through bit1 of the SCC’s Write Register 5.

Figure 2.  Driver Hardware Configuration

Figure 3.  SCC Clocking Scheme

HARDWARE CONFIGURATION (Continued)

Listing 1 (Reference Appendix A for Listings 1 through 4)
shows the assembler code for this SCC initialization. Note
that the SCC is treated as a peripheral by the Z80181’s
MPU. For example, an I/O write to the scc_cont (address
e8H) or to the scc_data (address e9H) is a write to the
SCC’s control and data registers, respectively. As shown
in Listing 1, the SCC is initialized by issuing I/O writes to the
pointer and then to the control registers in an alternating
fashion. It is therefore very important that all interrupts are
disabled during this initialization routine.

The SCC is initially reset through software before proceeding
to program the other write registers. A NOP is sufficient to
provide the four PCLKs required by the SCC recovery time
after a soft reset.  The SCC is programmed for SDLC mode.
The receive, transmit and external interrupts are all initially
disabled during this initialization. Each of these interrupt
sources are enabled at their proper times in the main
program. The SCC is programmed to include status
information in the vector that it places on the bus in
response to an interrupt acknowledge cycle (see Listing 4
of the SCC interrupt vector table for all the possible
sources).

Since SDLC is bit-oriented, the transmitter and receiver
are both programmed for 8 bits per character as required
by LLAP. Address filtering is implemented by setting the
Address Search Mode bit 2 on WR3. Setting this bit causes
messages with addresses not matching the address
programmed in WR6 and not matching the broadcast
address to be rejected. Values in WR10 presets the CRCs
to ones, sets the encoding to FM0 mode and makes
certain that transmission of flags occur during idle and
underrun conditions. WR11 is set so that the receive clock
is sourced by the DPLL output; the transmit clock is
sourced by the Baud Rate Generator output; /TRxC’s
output is from the BRG. The input to the BRG is from the /
RTxC.

The BRG’s time constant is loaded in WR13 and WR12 so
that the /RTxC’s 3.6864 MHz signal is divided by 16 in
order to obtain a 230.4 kHz signal for the transmitter clock.
WR14 makes certain that the DPLL is disabled before
choosing the clock source and operating mode. The DPLL
is enabled by issuing the Enter Search Mode in WR14.

TRANSMITTING A LLAP FRAME

Listing 2 shows the assembler code for subroutine txenq,
which sends an lapENQ frame on the line once the system
has determined that the line is quiet. Note that this routine
can easily be generalized to send any frame.

The first responsibility of txenq is to send the sync pulse
required by the CSMA/CA protocol. To do this, txenq sets
the /RTS pin active low, enabling the transmitter drivers,
and then sets it high again to disable them. In order to
satisfy the requirements of the CSMA/CA protocol, the
transmitter drivers must remain off for at least one bit time
(4.3 µsec) to guarantee that all the receivers see at least
one transition. Our routine satisfies this requirement
because the two ld instructions (7 T states each), the two
nop instructions (4 T states each) and the two "out"
instructions (11 T states each) required to set the /RTS line
high, take more than 4.3 µsec to execute on the 10 MHz
Z80181. The transmitter drivers must then remain off for at
least two bit times in order to ensure that all receivers lose
clock; again, the routine depends upon the time required
to execute the instructions before we turn the transmitter
drivers on again.

After sending the sync pulse and waiting the required
period of silence, txenq turns on the transmitter drivers to

send the frame. Now, the routine must wait while the SCC
sends out the leading flags. This takes 16 bit times, and
since the SCC does not tell when this has happened, the
transmit routine has no choice but to time this. Our routine
does this by calling bit time, which is discussed below.

When the two flags have been transmitted, the first data
byte is written to the data register of the SCC. Thereafter,
the routine polls the SCC status register, and when that
register shows the transmit buffer register is empty, the
routine sends the next data character. This polling method
can obviously be replaced by an interrupt routine that is
entered when the transmit buffer is empty or by setting up
the Z80181’s DMA to send characters on demand to the
SCC.

After the first data byte is transmitted, the txenq routine sets
the SCC to mark on idle so that the abort is sent when the
frame is over. This operation can be done any time after the
first data character has been placed in the transmit buffer
and before the trailing flag is shifted out. Txenq asserts this
mark on idle command after the first character is placed in
the transmit buffer so that LLAP has control and that no
issues of latency may arise (particularly in designs using
interrupt or DMA).

After the last data byte is written to the SCC, the transmit
routine must wait while the last data byte (the one that the
SCC had just sent to shifter), the two CRC bytes, one flag
byte and 12 to 18 bit times of marking are transmitted. This
total of 44 to 50 bit times is again timed by bittime. When
bittime indicates that enough time has elapsed, the
transmitter drivers are turned off.

Since our hardware includes connecting the output of the
baud rate generator to the input of counter/timer 1 on the
Z80181, that counter timer counts the bit times. The bit time
routine feeds an appropriate count value into the counter
and enables an interrupt routine to receive control when
the count expires. The interrupt routine ctc1int, shown in
Listing 4, sets the timeflag which the transmit routine polls.

Note that the call to bittime, the interrupt routine, the polling
code and the length of time it takes to write to the SCC
registers after a polling loop is exited, all take up a time that
can be a significant number of bits. In order to do these
timings accurately, calculate the number of PCLK cycles
required to execute these pieces of code and to adjust the
counter value that bittime requires. This adjustment is
dependent on the hardware configuration and on the exact
implementation details of the code.

Note, incidentally, that software must put the entire frame
into the transmit register, including the addresses. The
SCC does not generate addresses even when set in WR6.

RECEIVING LLAP FRAMES

In the experiments, the interrupt routines were used to
receive characters and to handle special conditions when
the frame is complete. Listing 3 shows the interrupt handlers
that are entered when the SCC receives a character and
when the SCC interrupts for a special condition (typically,
end of frame). As with transmission, it is obvious that we
could receive characters by polling the SCC (using up all
available CPU cycles) or by using DMA (using up very
few). It is estimated that the recint routine uses up about 1/
3 of the available 34 µsec (4.3 µsec x 8-bit times) cycles on
a 10 MHz processor.

The recint routine moves each character as it is received
from the SCC to a memory buffer and increments the buffer
pointer. The frame’s data length is checked to make
certain that the maximum allowable frame size is not
exceeded.

The spcond interrupt handler checks the status of the SCC
to find out what has happened. The presence on an
overrun condition or a CRC error is flagged as a receive
frame error.

The spcond routine decrements the receiver buffer address
by two to account for the two CRC bytes that are read from
the SCC before the special condition interrupt occurs.
Note that the SCC does not filter these CRC bytes, nor does
it filter the address byte. Everything received after the
leading flags and before the trailing flags appears in the
receive buffer.

One difficulty that arises in LLAP that was not addressed
here is that the receipt of a frame very often creates an
obligation to send a frame back to the sender within the
interframe gap, which is 200 µsecs. This difficulty is even
greater than it might appear. The 200 µsec gap starts when
the frame is received; it ends when the leading flags and
destination address of  the response are sent.  Start
sending the response soon enough to allow sending two
leading flags (plus a possible leading flag fragment) and
the first data character, and to allow for the 3-bit delay in the
SCC shifter. Therefore, start sending early enough to
transmit 35 bits before the interframe gap expires, or about
70 µsecs after you receive the frame.

CONCLUSIONS

The problems of sending the sync pulses, the timing of
transmission packets, and the problems associated with
the reception of packets as defined by LLAP are handled
by the Z80181 and its peripherals. It was demonstrated that
LLAP frames can be transmitted and received by using the
straight forward polling method and by using interrupt
routines. In a much busier environment where the processor
cannot strictly be an LLAP engine, other methods such as

using DMA in a fully interrupt driven environment must be
used. It was also demonstrated that severe CPU overhead
is used in setting up the sync pulses, timing out delays,
etc., before each LLAP frame. A modified SCC that transmits
and receives special LLAP frames helps in off loading
some of this overhead, hence freeing the CPU to do other
tasks.

APPENDIX A

Listing 1 - Assembler Code for SCC Initialization

LISTING 1

475  ;**************************************
476 ;subroutine to initialize scc registers
477  ;**************************************

000001e2 478 initscc:
000001e2 f3 479 di ;disable int while programming scc
000001e3 f5 480 push af
000001e4 c5 481 push bc
000001e5 e5 482 push hl

483
000001e6 3e09 484 ld a,09h ;WR9
000001e8 d3e8 485 out (scc_cont),a ;point to scc register
000001ea 3e80 486 ld a,80h ;channel  reset
000001ec d3e8 487 out (scc_cont),a ;scc register value
000001ee 00 488 nop ;delay needed after scc reset

489
490

000001ef 21Wwww 491 ld hl,scctable ;fetch start of scc init table
000001f2 492 scc1:
000001f2 7e 493 ld a,(hl) ;fetch register pointer value
000001f3 feff 494 cp 0ffh
000001f5 caWwww 495 jp z,finscc ;if reg a =0ffh then initscc finished
000001f8 d3e8 496 out (scc_cont),a
000001fa 23 497 inc hl
000001fb 7e 498 ld a,(hl)
000001fc d3e8 499 out (scc_cont),a
000001fe 23 500 inc hl
000001ff c3R000+01f2, 501 jp scc1 ;loop back

502
00000202 503 scctable:
00000202 04 504 db 04h ;WR4
00000203 20 505 db 00100000b ;sdlc uses 1x,sdlc mode,no parity

506
00000204 01 507 db 01h ;WR1
00000205 00 508 db 00h ;nothing,rx,tx and ext int disabled

509
00000206 02 510 db 02h ;WR2
00000207 00 511 db 00h ;vector base is 00h

512
00000208 03 513 db 03h ;WR3
00000209 cc 514 db 0cch ;rx 8b/char,rx crc enabled,address

515 ;search mode for adlc address filtering
516 ;rx disabled.
517

0000020a 05 518 db 05h ;WR5
0000020b 60 519 db 60h ;tx 8b/char, set rts to disable drivers

520
0000020c 06 521 db 06h ;WR6
0000020d 00 522 db 00h ;address field=’myaddress’ in main pgm

523
0000020e 07 524 db 07h ;WR7
0000020f 7e 525 db 7eh ;flag pattern

526
00000210 09 527 db 09h ;WR9
00000211 01 528 db 01h ;stat low, vis therefore vector returned

529 ;is a variable depending on the source
530 ;of the interrupt.

531
00000212 0a 532 db 0ah ;WR10
00000213 e0 533 db 0e0h ;crc preset to one,fm0, flag idle/undr

534
00000214 0b 535 db 0bh ;WR11
00000215 f6 536 db 0f6h ;rtxc=xtal,rxc=dpll,txc=brg,trxc=brg out

537
00000216 0c 538 db 0ch ;WR12
00000217 06 539 db 06h ;brg tc low,for 230.4kbps using rtxc=3.68MHz

540
00000218 0d 541 db 0dh ;WR13
00000219 00 542 db 00h ;brg tc high

543
0000021a 0e 544 db 0eh ;WR14
0000021b 60 545 db 60h ;disable dpll

546 ;no local loop back,brg source=rtxc
547

0000021c 0e 548 db 0eh ;WR14
0000021d c0 549 db 0c0h ;select fm mode

550 ;no local loop back,brg source=rtxc
551

0000021e 0e 552 db 0eh ;WR14
0000021f a0 553 db 0a0h ;dpll source=rtxc,

554 ;no local loop back,brg source=rtxc
555

00000220 0e 556 db 0eh ;WR14
00000221 20 557 db 20h ;enter search mode

558 ;no local loopback
559

00000222 0e 560 db 0eh ;WR14
00000223 01 561 db 01h ;null,no local loopback,enable the brg

562
00000224 03 563 db 03h ;WR3
00000225 cc 564 db 0cch ;rx 8b/c,enable rx crc,addrs src,rx disable

565
00000226 0f 566 db 0fh ;WR15
00000227 00 567 db 00h ;ext/stat not used

568
569 ;WR0

00000228 10 570 db 10h ;reset ext/stat once
00000229 10 571 db 10h ;reset ext/stat twice

572
0000022a 01 573 db 01h ;WR1
0000022b 00 574 db 00h ;disable all int sources

575
0000022c 09 576 db 09h ;WR9
0000022d 09 577 db 09h ;enable int
0000022e ff 578 db 0ffh ;finished

579
0000022f 580 finscc:
0000022f e1 581 pop hl ;
00000230 c1 582 pop bc ;
00000231 f1 583 pop af ;
00000232 c9 584 ret

585

Barbara E Lau
AN006401-0201



6-41

TECHNICAL  CONSIDERATIONS WHEN
IMPLEMENTING LOCALTALKZilog

AN971800400

LISTING 2

600 ;*****************************************
601 ;Subroutine to transmit the llapenq packet
602 ;*****************************************

00000244 603 txenq:
604

00000244 f5 605 push af ;save status and a reg
00000245 c5 606 push bc ;save
00000246 e5 607 push hl ;save

608 ;
00000247 f3 609 di ;make sure that

610 ;no interrupt routine
611 ;nor should interrupt
612 ;occur during
613 ;this subroutine.

00000248 3e03 614 ld a,03h
0000024a d3e8 615 out (scc_cont),a ;WR3
0000024c 3ecc 616 ld a,0cch
0000024e d3e8 617 out (scc_cont),a ;8b/char,rx crc

618 ;enable,addrs src
619 ;and rx disabled
620

00000250 3e0a 621 ld a,0ah ;select WR10
00000252 d3e8 622 out (scc_cont),a
00000254 3ee0 623 ld a,11100000b ;idle with flags
00000256 d3e8 624 out (scc_cont),a

625
626 ;****enable transmitter *****

00000258 3e05 627 ld a,05h ;select WR5
0000025a d3e8 628 out (scc_cont),a
0000025c 3e68 629 ld a,01101000b ;enable tx
0000025e d3e8 630 out (scc_cont),a

631 ;
632 ;
633 ;****enable rs-422 driver *****

00000260 3e05 634 ld a,05h ;select WR5
00000262 d3e8 635 out (scc_cont),a
00000264 3e6a 636 ld a,01101010b ;enable tx,
00000266 d3e8 637 out (scc_cont),a ;reset rts
00000268 00 638 nop
00000269 00 639 nop

640 ;nop’s needed to complete 4.3 usec
641 ;for 1 bit time enable of transmitter.
642 ;total delay=2*(7+11+4) T states at 10 MHZ
643 ;
644 ;****disable rs-422 driver for 2 bit times*****

0000026a 3e05 645 ld a,05h ;select WR5
0000026c d3e8 646 out (scc_cont),a
0000026e 3e68 647 ld a,01101000b ;enable tx, set rts
00000270 d3e8 648 out (scc_cont),a

649 ;
00000272 3e80 650 ld a,10000000b ;reset txcrc
00000274 d3e8 651 out (scc_cont),a
00000276 0601 652 ld b,01h ;delay count
00000278 653 csloop:
00000278 10fe 654 djnz csloop ;loop needed

655 ;to complete
656 ;8.6 usec  min.
657 ;or 2 bit times.
658 ;****enable rs-422 driver for llap transmission*****

0000027a 3e05 659 ld a,05h ;select WR5

0000027c d3e8 660 out (scc_cont),a
0000027e 3e6b 661 ld a,01101011b ;sdlc crc,

662 ;txcrc enable,
663 ;reset rts

00000280 d3e8 664 out (scc_cont),a
665
666  ;
667 ;**start counting out 2 flag character times **
668  ;
669 ;count 16 bit times
670 ;from the rs-422 enable
671 ;for 2 flags.
672 ;btdelay=subr delay+ctc1int+polling=8bits
673 ;16 bit times-btdelay=16-8=08h
674 ;

00000282 0e08 675 ld c,08h
00000284 cdWwww 676 call bittime ;bittime delay

677 ;is stored in reg.c
678 ;and bit1 of timflg
679 ;will indicate
680 ;count termination.
681

00000287682 l6: ;timer flag
00000287 3aWwww 683 ld a,(timflg) ;
0000028a cb4f 684 bit 1,a ;if bit1=1 then

685 ;count terminated
0000028c 28f9 686 jr z,l6 ;
0000028e cb8f 687 res 1,a ;reset timflg bit1
00000290 32Wwww 688 ld (timflg),a ;update timflg

689  ;
00000293 3e03 690 ld a,03h ;
00000295 d3e5 691 out (ctc1_cont),a ;disable int,

692 ;software reset
693 ;to kill the counter1

00000297 0602 694 ld b,02h ;2+1 bytes to transmitted
00000299 21Wwww 695 ld hl,txlapenq ;point to txlapenq buffer

696 ;send 1st byte
0000029c 7e 697 ld a,(hl) ;
0000029d d3e9 698 out (scc_data),a ;and send it
0000029f 23 699 inc hl ;point to the next byte

700  ;
000002a0 3ec0 701 ld a,0c0h ;reset eom latch command
000002a2 d3e8 702 out (scc_cont),a

703 ;
000002a4 f3 704 di ;disable all int
000002a5 3e0a 705 ld a,0ah ;select WR10
000002a7 d3e8 706 out (scc_cont),a

707 ;idle with 1’s
708 ;at the end of the frame

000002a9 3ee8 709 ld a,11101000b
000002ab d3e8 710 out (scc_cont),a

711 ;
000002ad 3e00 712 txq2: ld a,00h
000002af d3e8 713 out (scc_cont),a ;rr0
000002b1 dbe8 714 in a,(scc_cont) ;read rr0
000002b3 cb57 715 bit 2,a ;read tx buffer empty
000002b5 28f6 716 jr z,txq2 ;loop if zero
000002b7717 txq1:
000002b7 7e 718 ld a,(hl) ;
000002b8 d3e9 719 out (scc_data),a ;and send it
000002ba 23 720 inc hl ;point to the next byte

721

000002bb 10f0 722 djnz txq2 ;loop until all
723 ;bytes have been
724 ;transmitted.
725
726

000002bd 3e28 727 ld a,028h ;reset tx int pending
000002bf d3e8 728 out (scc_cont),a

729 ;note:tx buffer
730 ;empty happens as tx
731 ;shifter is loaded.
732  ;
733 ;count= last byte+
734 ;crc+flag+12bit times-btdelay
735 ;btdelay=subr delay+ctc1int+polling=8bits
736 ;8+16+8+12-8=36=24h

000002c1 0e24 737 ld c,24h
000002c3 cdWwww 738 call bittime ;bittime delay

739 ;is stored in reg.c
740  ;

000002c6741 l7: ;timer flag
000002c6 3aWwww 742 ld a,(timflg) ;
000002c9 cb4f 743 bit 1,a ;if bit1=1 then count finish
000002cb 28f9 744 jr z,l7 ;
000002cd cb8f 745 res 1,a ;reset timflg bit1
000002cf 32Wwww 746 ld (timflg),a ;update timflg

747 ;
000002d2 3e03 748 ld a,03h ;
000002d4 d3e5 749 out (ctc1_cont),a ;disable int,software reset

750 ;to kill counter
751 ;****disable rs-422 driver after 12 to 18 1’s*****

000002d6 3e05 752 ld a,05h ;select WR5
000002d8 d3e8 753 out (scc_cont),a
000002da 3e60 754 ld a,01100000b ;disable tx, set rts
000002dc d3e8 755 out (scc_cont),a

756
000002de 3e03 757 ld a,03h
000002e0 d3e8 758 out (scc_cont),a ;WR3
000002e2 3ecd 759 ld a,0cdh
000002e4 d3e8 760 out (scc_cont),a ;8b/char,rx crc enabled,

761 ;address search and rx enabled
762
763 ;*************************************
764
765 ;count for the interframe gap
766 ;of 200 usec or 46 bit times.
767 ;btdelay=subr delay+ctc1int+polling=8bits
768 ;46 - btdelay=46-8=26h
769 ;note that timflg will be polled in
770 ;the main routine.
771 ;

000002e6 0e26 772 ld c,26h
000002e8 cdWwww 773 call bittime

774 ;
775 ;bittime delay is stored in reg.c
776 ;*************************************

000002eb e1 777 pop hl ;restore
000002ec c1 778 pop bc ;restore
000002ed f1 779 pop af ;restore status and a reg
000002ee c9 780 ret

781
782
783

;**********************************************
784 ;subroutine to time out bit time 4.3 usec per bit
785 ;register c contains the number of bits to be

counted down
786 ;**********************************************

000002ef 787 bittime:
000002ef f5 788 push af ;save status and a reg
000002f0 c5 789 push bc ;save
000002f1 e5 790 push hl ;save

791 ;
000002f2 3ed2 792 ld a,0d2h ;ctc1 int vector
000002f4 d3e5 793 out (ctc1_cont),a

794
000002f6 3edf 795 ld a,11011111b ;
000002f8 d3e5 796 out (ctc1_cont),a ;enable int

797 ;select counter mode
798 ;clk/trg edge starts with rising edg
799 ;time constant follows
800 ;software reset

000002fa 79 801 ld a,c ;reg c contains the number of bits
000002fb d3e5 802 out (ctc1_cont),a ;load the number of bits to be counted

803 ;**
000002fd e1 804 pop hl ;restore
000002fe c1 805 pop bc ;restore
000002ff f1 806 pop af ;restate status and a reg
00000300 fb 807 ei
00000301 c9 808 ret

809

LISTING 3

1131 ;****************************
1132 ;receive int service routine.
1133 ;****************************
1134 ;save received character in receiver buffer

pointed
1135 ;to by rxpointer
1136

0000044d 1137recint:
0000044d f5 1138 push af ;save af
0000044e d5 1139 push de
0000044f e5 1140 push hl
00000450 dbe9 1141 in a,(scc_data) ;read scc data
00000452 2aWwww 1142 ld hl,(rxpointer) ;
00000455 77 1143 ld (hl),a ;save it
00000456 23 1144 inc hl ;update pointer
00000457 22Www 1145 ld (rxpointer),hl ;
0000045a ed5bWwww 1146 ld de,(rxbufend) ;end of rx buffer
0000045e af 1147 xor a ;reset cy
0000045f ed52 1148 sbc hl,de ;
00000461 c2Wwww 1149 jp nz,recexit ;if not zero,then receive

byte length is ok
00000464 21Wwww 1150 ld hl,recerrflg ;
00000467 cbc6 1151 set 0,(hl) ;set bit0=1 maxfrmflg to indicate error

1152 ;because of max frame
size exceeded.

00000469 1153recexit:
00000469 3e38 1154 ld a,038h
0000046b d3e8 1155 out (scc_cont),a ;reset highest ius
0000046d e1 1156 pop hl
0000046e d1 1157 pop de
0000046f f1 1158 pop af ;restore af
00000470 fb 1159 ei ;enable int
00000471 c9 1160 ret ;return from int

1161 ;note ret and not reti is used for scc
1162 ;interrupts on the z80181.
1163
1164 ;*****************************************
1165 ;special receive interrupt service routine
1166 ;*****************************************
1167; “parity is special condition” bit is off.
1168; special conditions are eof or rx overrun error.
1169; crc error flag is valid only if eof is valid.
1170; if frame is ok then recerrflg bit1=0, otherwise

bit1=1.
1171

00000472 1172 spcond:
00000472 f5 1173 push af ;save af reg
00000473 c5 1174 push bc ;
00000474 e5 1175 push hl;

1176
00000475 3e01 1177 ld a,01h
00000477 d3e8 1178 out (scc_cont),a ;read rr1
00000479 dbe8 1179 in a,(scc_cont)
0000047b e660 1180 and 01100000b  ;check bit6 (crc) or bit5 (overrun)
0000047d caWwww 1181 jp z,ok ;

1182 ;
00000480 21Wwww 1183 ld hl,recerrflg ;fetch receive error flag
00000483 cbce 1184 set 1,(hl) ;set bit1=1 for frame not ok
00000485 c3Wwww 1185 jp crc_exit

00000488 1186 ok:
00000488 21Wwww 1187 ld hl,recerrflg ;fetch receive error flag
0000048b cb8 1188 res 1,(hl) ;set bit1=0 for frame  ok

1189
1190

0000048d 1191 crc_exit:
0000048d dbe9 1192 in a,(scc_data) ;read 2nd crc (debug only) and

scrap
0000048f 2aWwww 1193 ld hl,(rxpointer) ;load pointer
00000492 2b 1194 dec hl ;adjust rx buff ptr for crc1
00000493 2b 1195 dec hl ;adjust rx buff ptr for crc2
00000494 22Wwww 1196 ld (rxpointer),hl ;

1197
00000497 1198 spexit:
00000497 3e38 1199 ld a,038h
00000499 d3e8 1200 out (scc_cont),a ;reset highest ius

1201
0000049b e1 1202 pop hl ;restore hl
0000049c c1 1203 pop bc ;restore be
0000049d f1 1204 pop af ;restore af
0000049e fb 1205 ei ;enable int
0000049f c9 1206 ret ;return from int

1207

LISTING 4

1306
1307 ;****************************************
1308 ;ctc1 timer int handler
1309 ;****************************************

00000509 1310 ctc1int:
1311 ;ctc1 is programmed in counter mode.
1312 ;external trigger edges is provided by
1313 ;/trxc pin at intervals of 4.3 usec.
1314 ;bit1 of timflg is set when count is terminated.

00000509 f5 1315 push af
0000050a c5 1316 push bc
0000050b e5 1317 push hl

1318 ;** update the timing flag **
0000050c 21Wwww 1319 ld hl,timflg
0000050f 7e 1320 ld a,(hl) ;get recent timflg
00000510 cbcf 1321 set 1,a ;bit1=1 after count is over
00000512 77 1322 ld (hl),a ;update the timflg
00000513 e1 1323 pop hl
00000514 c1 1324 pop bc
00000515 f1 1325 pop af
00000516 fb 1326 ei
00000517 ed4d 1327 reti

1328
1329
1330
1331 ;**********************************
1332 ;interrupt vector table for the scc
1333 ;**********************************
1334 ;the status of the interrupt source will affect
1335 ;the interrupt vector. The interrupt handler’s
1336 ;address are set in a block, as below.

00000a00 1337 org sdlc + 0a00h
00000a00 1338 sccvect:

1339 if scc_a
00000a00 1340 .block 8 ;reserve vector for other ch

1341 endif
00000a08 R000+03e9, 1342 dw txint ;tx int
00000a0a R000+04c8, 1343 dw ext_stat ;ext/stat int
00000a0c R000+0433, 1344 dw recint ;rx char int
00000a0e R000+0454, 1345 dw spcond ;sp rec cond int

1346
1347 if not scc_a

00000a10 1348 .block 8 ;reserve vector for other ch
1349 endif
1350

00000a18 1351 temp: .block 1
1352
1353
1354 ;**********************************
1355 ;interrupt vector table for the ctc
1356 ;**********************************

00000ad0 1357 org 0ad0h
00000ad0 R000+04d8, 1358 dw ctc0int ;reserved for ctc0 int routine
00000ad2 1359 org 0ad2h
00000ad2 R000+0509, 1360 dw ctc1int ;reserved for ctc1 int routine

1361
1362 ;************************
1363 ;receive buffer area
1364 ;************************

00001000 1365 org  1000h
00001000 1366 rx_buff: .block length

1367
1388
1389 ;************************
1390 ;transmitter buffer area
1391 ;************************

0000b000 1392 org 0b000h
1398 ;
1399 ;********************************
1400 ;transmit llap enq packet (3bytes)
1401 ;********************************

0000b258 ff 1402 txlapenq: db 0ffh ;broadcast id
0000b259 1403 myaddress .block 1 ;guess at myaddress
0000b25a 81 1404 db 81h ;llap enq type

1405 ;

APPENDIX B

12 to 18 1's at the end of an LLAP frame

APPENDIX B (Continued)

CSMA/CA before an LLAP frame

An LLAP Frame

Zilog’s products are not authorized for use as critical compo-
nents in life support devices or systems unless a specific written
agreement pertaining to such intended use is executed between
the customer and Zilog prior to use. Life support devices or
systems are those which are intended for surgical implantation
into the body, or which sustains life whose failure to perform,
when properly used in accordance with instructions for use
provided in the labeling, can be reasonably expected to result in
significant injury to the user.

